/*
* Licensed under the MIT License
* 
* Copyright (c) 2010 Kay Kreuning
* 
* Permission is hereby granted, free of charge, to any person obtaining a copy
* of this software and associated documentation files (the "Software"), to deal
* in the Software without restriction, including without limitation the rights
* to use, copy, modify, merge, publish, distribute, sublicense, and/or sell
* copies of the Software, and to permit persons to whom the Software is
* furnished to do so, subject to the following conditions:
* 
* The above copyright notice and this permission notice shall be included in
* all copies or substantial portions of the Software.
* 
* THE SOFTWARE IS PROVIDED "AS IS", WITHOUT WARRANTY OF ANY KIND, EXPRESS OR
* IMPLIED, INCLUDING BUT NOT LIMITED TO THE WARRANTIES OF MERCHANTABILITY,
* FITNESS FOR A PARTICULAR PURPOSE AND NONINFRINGEMENT. IN NO EVENT SHALL THE
* AUTHORS OR COPYRIGHT HOLDERS BE LIABLE FOR ANY CLAIM, DAMAGES OR OTHER
* LIABILITY, WHETHER IN AN ACTION OF CONTRACT, TORT OR OTHERWISE, ARISING FROM,
* OUT OF OR IN CONNECTION WITH THE SOFTWARE OR THE USE OR OTHER DEALINGS IN
* THE SOFTWARE.
* 
* http://code.google.com/p/as3hyvesapi
*/
package nl.hyves.api.sessions
{
	import flash.external.ExternalInterface;
	import flash.net.URLRequest;
	import flash.net.navigateToURL;
	
	import nl.hyves.api.delegates.IAPICallDelegate;
	import nl.hyves.api.methods.PublicMethods;
	import nl.hyves.api.utils.AuthUtil;
	
	import org.osflash.signals.ISignal;
	import org.osflash.signals.Signal;
	
	public class PopupSession implements ISession
	{
		private static const AUTHORIZE_ENDPOINT:String = "http://www.hyves.nl/api/authorize/";
		
		public static const EXPIRATIONTYPE_DEFAULT:String = "default";
		public static const EXPIRATIONTYPE_INFINITE:String = "infinite";
		public static const EXPIRATIONTYPE_USER:String = "user";
		
		private var _accesstokenDispatcher:Signal;
		private var callbackURL:String;
		private var methods:Array;
		private var expirationType:String;
		private var callMethod:Function;
		
		public function get accesstokenDispatcher():ISignal
		{
			return _accesstokenDispatcher;
		}
		
		public function PopupSession(callbackURL:String, methods:Array, expirationType:String = EXPIRATIONTYPE_DEFAULT)
		{
			_accesstokenDispatcher = new Signal(Accesstoken);
			
			this.callbackURL = callbackURL;
			this.methods = methods;
			this.expirationType = expirationType;
		}
		
		public function connect(callMethod:Function):void
		{
			this.callMethod = callMethod;
			
			var parameters:Object = {
				methods: methods.join(","),
				expirationtype: expirationType
			};
			
			var delegate:IAPICallDelegate = callMethod.apply(null, [PublicMethods.auth_requesttoken, parameters]);
			
			delegate.resultHandler.addOnce(onRequesttoken);
		}
		
		public function exchange():void
		{
			var delegate:IAPICallDelegate = callMethod.apply(null, [PublicMethods.auth_accesstoken]);
			
			delegate.resultHandler.addOnce(onAccesstokenComplete);
		}
		
		private function onRequesttoken(result:Object):void
		{
			var accesstoken:Accesstoken = AuthUtil.extractAccesstoken(result);
			
			_accesstokenDispatcher.dispatch(accesstoken);
			
			var url:String = AUTHORIZE_ENDPOINT + "?oauth_token=" + encodeURIComponent(accesstoken.token) +
				"&oauth_callback=" + encodeURIComponent(callbackURL);
			
			if (ExternalInterface.available && ExternalInterface.call("OAuthBridge.available"))
			{
				ExternalInterface.addCallback("setVerifier", onRequesttokenAuthorized);
				ExternalInterface.call("OAuthBridge.callAuthorize", url);
			}
			else
			{
				// Resort to a native Flash 'popup'
				navigateToURL(new URLRequest(url));
			}
		}
		
		private function onRequesttokenAuthorized(response:String):void
		{
			var token:String = decodeURIComponent(response.match(/(?<=oauth_token).*(?=&oauth_verifier)/g)[0]);
			var verifier:String = decodeURIComponent(response.match(/(?<=oauth_verifier=)\w+/g)[0]);
			
			var delegate:IAPICallDelegate = callMethod.apply(null, [PublicMethods.auth_accesstoken, {oauth_token: token, oauth_verifier: verifier}]);
			
			delegate.resultHandler.addOnce(onAccesstokenComplete);
		}
		
		private function onAccesstokenComplete(result:Object):void
		{
			_accesstokenDispatcher.dispatch(AuthUtil.extractAccesstoken(result));
		}
	}
}